Freier Fall aus der Stratosphäre (Christian Spreitzer, Juli 2013):

Luftdichte als Funktion der Höhe in km (Daten aus US Standard Atmosphere 1976):

```
ddata := {{0, 1.225}, {1, 1.1117}, {2, 1.0066}, {3, 0.90925}, {4, 0.81935},
  {5, 0.73643}, {6, 0.66011}, {7, 0.59002}, {8, 0.52579}, {9, 0.46706},
  {10, 0.41351}, {11, 0.3648}, {12, 0.31194}, {13, 0.2666}, {14, 0.22786},
  {15, 0.19476}, {16, 0.16647}, {17, 0.1423}, {18, 0.12165}, {19, 0.104},
  {20, 0.08891}, {21, 0.075715}, {22, 0.06451}, {23, 0.055006}, {24, 0.04938},
  {25, 0.040084}, {26, 0.034257}, {27, 0.029298}, {28, 0.025076},
  {29, 0.021478}, {30, 0.01841}, {31, 0.015792}, {32, 0.013555},
  {33, 0.011573}, {34, 0.0098874}, {35, 0.0084634}, {36, 0.0072579},
  {37, 0.0062355}, {38, 0.0053666}, {39, 0.0046268}, {40, 0.0039957}}
```

```
ListPlot[ddata]
```



Zum Vergleich: Exponentielle Abnahme der Luftdichte in einer Atmosphäre konstanter Temperatur:

```
df[h_] := 1.225 * Exp[-h / 8]
```

```
Plot[df[h], {h, 0, 40}, Epilog → Map[Point, ddata]]
```



Interpolation der Atmosphärendichtedaten:

```
dint = Interpolation[ddata]
```

```
InterpolatingFunction[{{0., 40.}}, <>]
```

```
Plot[dint[h], {h, 0, 40}, Epilog → Map[Point, ddata]]
```



Schallgeschwindigkeit als Funktion der Höhe in km (Daten aus US Standard Atmosphere 1976):

```
sdata := {{0, 340.29}, {1, 336.43}, {2, 332.53}, {3, 328.58}, {4, 324.59},
  {5, 320.55}, {6, 316.45}, {7, 312.31}, {8, 308.11}, {9, 303.85}, {10, 299.53},
  {11, 295.15}, {12, 295.07}, {13, 295.07}, {14, 295.07}, {15, 295.07},
  {16, 295.07}, {17, 295.07}, {18, 295.07}, {19, 295.07}, {20, 295.07},
  {21, 295.7}, {22, 296.38}, {23, 297.05}, {24, 297.72}, {25, 298.39},
  {26, 299.06}, {27, 299.72}, {28, 300.39}, {29, 301.05}, {30, 301.71},
  {31, 302.37}, {32, 303.02}, {33, 304.67}, {34, 306.49}, {35, 308.3},
  {36, 310.1}, {37, 311.89}, {38, 313.67}, {39, 315.43}, {40, 317.19}}
```

```
sint = Interpolation[sdata]
```

```
InterpolatingFunction[{{0., 40.}}, <>]
```

```
Plot[sint[h], {h, 0, 40}, Epilog → Map[Point, sdata]]
```



Strömungswiderstandsbeiwert als Funktion der Machzahl (Daten aus Jandir et al.):

Drag coefficient as a function of Mach number for a sphere:

```
DS[m_] := Piecewise[{{0.45 * m^2 + 0.424, m < 0.722},
    {2.1 * Exp[-1.2 * (m + 0.35)] - 8.9 * Exp[-2.2 * (m + 0.35)] + 0.92, m ≥ 0.722}}]
```

```
Plot[DS[m], {m, 0, 3}]
```




```
DS[0]
```

```
0.424
```

Dimensionless function modeling the flat spin period of F. B. (Funktion anhand Videoaufnahmen des Sprungs modelliert):

```
FSRR[r_] := Piecewise[{{1, r < 50}, {1 + (r - 50) / 15, 50 ≤ r < 65}, {2, 55 ≤ r}}]
```

```
Plot[FSRR[t], {t, 10, 100}]
```



Erdbeschleunigung in großer Höhe:

```
greal[h_] := 9.80665 * (6 371 000 / (6 371 000 + h)) ^ 2
```

Kalibrierung des Modells im Geschwindigkeitsmaximum:

```
hkal := 27 833
```

```
vkal := 1357.6 / 3.6
```

```
b1 := greal[hkal] / (df[hkal / 1000] * vkal^2)
```

```
b2 := greal[hkal] / (dint[hkal / 1000] * vkal^2)
```

Originale Daten vom Stratosphärensprung:

- **Zeit-Geschwindigkeit:**

```
speeddata := {{34, 1115 / 3.6}, {50, 1357.6 / 3.6},
  {64, 1043 / 3.6}, {180, 285 / 3.6}, {260, 191.5 / 3.6}}
```

- **Zeit-Höhe:**

```
heightdata := {{34, 33 446}, {50, 27 833}, {64, 22 960.7}, {180, 7619}}
```

Differentialgleichung und numerische Lösung (Geschwindigkeit in m/s vs. Zeit in s) mit expontiell abnehmender Luftdichte:

```
s1 := NDSolve[{h''[t] == -greal[h[t]] + df[h[t] / 1000] * h'[t] * h'[t] * b1,
    h[0] == 38 969, h'[0] == 0}, h, {t, 0, 1000}];
```

```
Plot[Evaluate[-h'[t] /. s1], {t, 0, 300},
 PlotRange → All, Epilog → Map[Point, speeddata]]
```



Differentialgleichung und numerische Lösung (Geschwindigkeit in m/s vs. Zeit in s) mit dem Modell der US Standard Atmosphere:

```
s2 := NDSolve[{h''[t] == -greal[h[t]] + dint[h[t] / 1000] * h'[t] * h'[t] * b2,
    h[0] == 38 969, h'[0] == 0}, h, {t, 0, 1000}];
```

```
Plot[Evaluate[-h'[t] /. s2], {t, 0, 300},
 PlotRange → All, Epilog → Map[Point, speeddata]]
```

InterpolatingFunction::dmval :
Input value {−0.122947} lies outside the range of data in the interpolating function. Extrapolation will be used. ≫

InterpolatingFunction::dmval :
Input value {−0.122947} lies outside the range of data in the interpolating function. Extrapolation will be used. ≫

InterpolatingFunction::dmval :
Input value {−0.122947} lies outside the range of data in the interpolating function. Extrapolation will be used. ≫

General::stop : Further output of InterpolatingFunction::dmval will be suppressed during this calculation. ≫



```
Show[%58, ImageSize → Full]
```



Differentialgleichung und numerische Lösung mit dem Modell der US Standard Atmosphere und Berücksichtigung der Schallbarriere:

```
s3 := NDSolve[{h''[t] == -greal[h[t]] +
      dint[h[t] / 1000] * h'[t] * h'[t] * b2 * DS[Abs[h'[t]] / sint[h[t] / 1000]],
    h[0] == 38 969, h'[0] == 0}, h, {t, 0, 1000}];
```

```
Plot[Evaluate[-h'[t] /. s3], {t, 0, 300},
 PlotRange → All, Epilog → Map[Point, speeddata]]
```

InterpolatingFunction::dmval : Input value {−0.0123249} lies
outside the range of data in the interpolating function. Extrapolation will be used. ≫

InterpolatingFunction::dmval : Input value {−0.0123249} lies
outside the range of data in the interpolating function. Extrapolation will be used. ≫

InterpolatingFunction::dmval : Input value {−0.0123249} lies
outside the range of data in the interpolating function. Extrapolation will be used. ≫

General::stop : Further output of InterpolatingFunction::dmval will be suppressed during this calculation. ≫




```
Show[%42, ImageSize → Full]
```



HIER WURDE (ABGESEHEN VON DER KALIBRIERUNG DES WIDERSTANDSBEIWERTS IM GESCHWINDIGKEITSMAXIMUM) KEINERLEI PARAMETER ANGEPASST - DIES IST DIE THEORETISCH VORHERGESAGTE KURVE (BEI UNVERÄNDERTER KÖRPERHALTUNG WÄHREND DES SPRUNGS, SO WIE BEIM GESCHWINDIGKEITSMAXIMUM).

```
Plot[Evaluate[h[t] /. s3], {t, 0, 300},
 PlotRange → All, Epilog → Map[Point, heightdata]]
```

InterpolatingFunction::dmval : Input value {−0.0123249} lies outside the range of data in the interpolating function. Extrapolation will be used. ≫

InterpolatingFunction::dmval : Input value {−0.0123249} lies outside the range of data in the interpolating function. Extrapolation will be used. ≫

InterpolatingFunction::dmval : Input value {−0.0123249} lies outside the range of data in the interpolating function. Extrapolation will be used. ≫

General::stop : Further output of InterpolatingFunction::dmval will be suppressed during this calculation. ≫




```
Show[%46, ImageSize → Full]
```



Zusätzliche Berücksichtigung der Trudelphase F.B.s:

```
s4 :=
  NDSolve[{h''[t] == -greal[h[t]] + dint[h[t] / 1000] * h'[t] * h'[t] * b2 * FSRR[t] *
       DS[Abs[h'[t]] / sint[h[t] / 1000]],
    h[0] == 38 969, h'[0] == 0}, h, {t, 0, 1000}];
```

```
Plot[Evaluate[-h'[t] /. s4], {t, 0, 300},
 PlotRange → All, Epilog → Map[Point, speeddata]]
```

InterpolatingFunction::dmval :
Input value {−0.020797} lies outside the range of data in the interpolating function. Extrapolation will be used. ≫

InterpolatingFunction::dmval :
Input value {−0.020797} lies outside the range of data in the interpolating function. Extrapolation will be used. ≫

InterpolatingFunction::dmval :
Input value {−0.020797} lies outside the range of data in the interpolating function. Extrapolation will be used. ≫

General::stop : Further output of InterpolatingFunction::dmval will be suppressed during this calculation. ≫



```
Show[%82, ImageSize → Full]
```




```
Plot[Evaluate[h[t] /. s4], {t, 0, 300},
 PlotRange → All, Epilog → Map[Point, heightdata]]
```

InterpolatingFunction::dmval :
Input value {−0.020797} lies outside the range of data in the interpolating function. Extrapolation will be used. ≫

InterpolatingFunction::dmval :
Input value {−0.020797} lies outside the range of data in the interpolating function. Extrapolation will be used. ≫

InterpolatingFunction::dmval :
Input value {−0.020797} lies outside the range of data in the interpolating function. Extrapolation will be used. ≫

General::stop : Further output of InterpolatingFunction::dmval will be suppressed during this calculation. ≫



```
Show[%84, ImageSize → Full]
```